NPN_Ftau_15uu_TF




```
.OPTIONS    GMIN=1e-15   METHOD=gear   ABSTOL=1e-15  TEMP=27 srcsteps = 1  gminsteps = 1
VCC     C     0     DC     5V
QN1     C     B     E      NPNV
R1      0     B            1
I1      E     0     DC     100u    AC    .01u
.control
echo              "15um^2 Ftau vs IE and TF"
setplot           new
set NameList =    ( TF_1p TF_2p TF_4p )
compose TfVals   values  1p 2p 4p
compose IbiasVals values 10u 100u 1m 10m 30m 60m 100m
settype current   IbiasVals
let NoOfTf   =   length(TfVals)
let NoOfIbias =   length(IbiasVals)
begin
unset             interrupt
* ================Loop_CJE============
let j =           1
while             ( j <= NoOfTf )
let tf =         TfVals[j-1]
altermod          npnv tf = $&tf
echo              "=>Ibias = Ftau@TF=$&tf"
set thisName =    $NameList[$&j]
let $thisName =   0*vector(NoOfIbias)
* ================Loop_Ibias==============
let k =           1
while             ( k <= NoOfIbias )
let Ibias =       IbiasVals[k-1]
alter I1 dc =     $&Ibias
alter I1 ac =     .01u
ac                dec 30 .0001GHz 100GHz
* ================Find_Ftau============= Use
let g =           (-db(b) - 160)
let f =           3.16*mag(frequency[ sortorder( abs(g-10.0) )[0] ] )
echo              "$&unknown.Ibias   $&f"
let               unknown.{$thisName}[unknown.k-1] = f
destroy
if                ($?interrupt)
goto              bail
endif
let k =           k + 1
endwhile
let j =           j + 1
endwhile
setscale          IbiasVals
plot              $NameList loglog title "15um^2 Ftau vs IE and TF"
label             bail
echo             "Done."
end
.endc
```

```
.MODEL  NPNV       NPN(
+ IS=10.15e-18     NF=1.0        BF=89         VAF=107      IKF=.4
+                  NR=1.006      BR=0.4822     VAR=4.286    IKR=0.0002472
+ ISE=9.15E-17     NE=2
+ ISC=1E-21        NC=2
+ RB=12            RBM=3                                    IRB=.1
+ RE=2             RC=33
+ CJE=37e-15       VJE=0.75      MJE=0.35
+ CJC=45E-15       VJC=0.6399    MJC=0.3531
+ CJS=2.939E-15    VJS=0.3488    MJS=0.1813   XCJC=0.4201
+ TF=1e-12         XTF=1         VTF=1.5      ITF=.8
+ TR=5.9e-9        FC=0.5        PTF=30
+ KF=1.000E-16     AF=1
+ XTB=2            EG=1.11       XTI=5        TNOM=25       )

.end
```

==========END_OF_SPICE==========

To Covert PDF to plain text click below
http://www.fileformat.info/convert/doc/pdf2txt.htm




Graph 94 - unknown419: 15um^2 Ftau vs IE and TF
Hz
TF_1p
TF_4p
TF_2p
10^11
10^10
10^9
10^8
frequency
10^-5
10^-4
10^-3
0.01
0.1
IbiasVals A

=